
Introduction

In my readings on computer graphics and particularly computational geometry, I found several scattered references to skip lists. They were said to provide very fast access to sorted information. But I had never seen a detailed description. Just recently, I stumbled on a Java implementation by Thomas Wenger1. Why not reimplement it in Smalltalk to get a better understanding, I thought?

Skip Lists

Skip lists are a form of sorted linked lists with an arbitrary number of “next” pointers that can skip ahead to elements far away in the list. Another way to look at it is to view it as a multi-resolution list (see Figure 1). For example, at a gross resolution (the top level), the list contains only elements “c” and “f”. If we are searching for these elements, we can find them quickly. On the other hand, if we are searching for “e”, we know that it must lie somewhere between “c” and “f”. Hence we can drop to a higher resolution sublist between “c” and “f” and immediately locate “e”. We can repeat this strategy dropping to higher and higher resolutions until we reach the highest resolution sublist at the bottom level. 

Figure 1   A typical skip list.



We can see that a skip list can be manipulated as a standard list by walking the bottom level (the highest resolution level). Also, if an element exists at a particular resolution, it also exists at all higher resolutions. 



When a new item is added to the list, we have to decide which resolution level it should belong to. Items that exist only at the lowest resolution, like “d”, only need to keep track of one next pointer. It can be described as a height 1 item. Item “e”, by contrast, is a height 2 item and item “f” is a height 3 item. When a new item at height h is introduced into the list, h predecessor pointers must be redirected at the new item and the new item must use these original values for its next pointers.



In order to ensure that we get good average search times for a skip list, William Pugh2 has shown that it is enough to pick the height randomly in such a matter there there is a higher probability of picking a height 1 than a height 2, a higher probability of picking a height 2 than a height 3, etc.

The Wenger Implementation in Smalltalk

Listing 1 (containing classes SkipListElement and SkipList) is a relatively straightforward port of the Wenger implementation in Java. The three most important methods are insert:value: (for inserting a key/value pair), search: (for finding the value associated with a key), and remove: (for deleting a particular key; the original was called delete: but Smalltalk subscribes to the remove: terminology). Skip lists are intended to be used as shown below.

 

	example1

		"SkipList example1"

		"SkipList example1 search: 50"

		"SkipList example1 remove: 50"

		| list |

		list := SkipList new: 50.

		#(10 50 20 40 5 100) do: [:integer |

			list insert: integer value: integer printString].

		^list



The most obvious changes we made was (1) to switch from 0-based indexing to 1-based indexing and (2) abstracting search methods for use by the search, insert, and remove methods. The most striking deficiency of the implementation was that it was designed to work only for integer keys greater than 0 and less than the maximum allowed Java integer. These two values are maintained by the header and trailer skip list elements. The fact that the header and trailer are skip list elements (rather than, say, just arrays) eliminates special case code for dealing with the list extremities—a nice idea.

Extending the Skip List Implementation

The most obvious extensions we wanted to make (see Listing 2) was to generalize the implementation for keys that are arbitrary magnitudes and to provide a more Smalltalk-like protocol via messages like at:, at:put:, do:, and printOn:. But before we get to that, let’s consider the minor changes. 



In particular, we changed the “level” terminology that was used throughout to “height”. We felt that a reader ought to have some intuition when talking about an element at level 3. Is it 3 from the top or three from the bottom. The word “level” itself provides no hints. A word like “depth” would suggest that it means from the top; “height” would suggest from the bottom. Since the bottom is invariant (tops vary continuously), we chose “height”. A second small change we made was to make the “nexts” array in the skip list element private (Wenger actually called this “forward”). To deal with that, we introduce methods “nextAt: height” and “nextAt: height put: anObject” and a special case “next” which means “nextAt: 1”. COMMENTTOEDITORPLEASEDONOTMOVETHEPERIODINTOTHEQUOTE



We also made a few innocuous changes like storing the trailer in addition to the header and not storing the maximum height (this can be computed by asking the header or trailer for its height). The maximum height is computed from an estimate of the size of the skip list when the list is created. The height of the list is the maximum of the heights of its elements and starts off at 1 when the list is empty. The list height is used when computing a random height for a new element. If the current height is h, we permit new elements with heights between 1 and h+1 but not h+2 or higher (although it would still work if we permitted it). 

Dealing with Arbitrary Magnitudes

To generalize the integer keys to arbitrary magnitudes, we had to decide what to do with the keys in the header and trailer. These keys are currently used when searching in methods “atOrPast: aMagnitude” and “predecessorsOf: aMagnitude”. Both methods have a loop of the form



		[aMagnitude > next key] whileTrue: [

			“move further right”]



If “next” is one of the sentinels (the trailer or header), we could, for example, special case it. By using nil as a sentinel key, we might for example write the following instead.



		[next notNil and: [aMagnitude > next key]] whileTrue: [

			“move further right”]



Another alternative is to make the sentinels themselves understand the compare operator. To do that, we created two subclasses of SkipListElement called SkipListHeader and SkipListTrailer. We provided three implementations of < as follows:



	In SkipListElement

	< aMagnitude

		^self key < aMagnitude



	In SkipListHeader

	< aMagnitude

		^true



	In SkipListTrailer

	< aMagnitude

		^false



Then we rewrote the above search loop as follows (we have to be careful to ensure that the skip list element is the receiver, not the magnitude):



		[next < aMagnitude] whileTrue: [

			“move further right”]



Even though we had to create two new classes, we settled on this design as the cleanest approach. The classes really are private to the implementation and should not concern the user.

Dealing with a more Smalltalk-like Protocol

When we introduced the at:, at:ifAbsent:, and at:put: protocol, it was clear that we had to generalize the search and remove methods to provide an error block.



	search: aMagnitude ifAbsent: aBlock

		…

	remove: aMagnitude ifAbsent: aBlock

		…



The at:, at:ifAbsent:, and at:put: methods could then be implemented as synonyms for search:, search:ifAbsent: , and insert:value: respectively.



To provide sequencing methods do: and keysAndValuesDo:, we merely needed to sequence through the elements at height 1. But this suggested that we might wish to sequence through the elements at any height. So we ended up implementing a looping construct that sequences through the skip list elements at a specified height as follows:



	atHeight: height listElementDo: aBlock

		"Sequence through the list at the given height and supply 	the skip list elements (exclude the header/trailer)."

		| current next |

		current := header nextAt: height.

		next := current nextAt: height.

		[next notNil] whileTrue: [

			aBlock value: current.

			current := next. next := current nextAt: height].



To provide the list values, we could then implement the following:



	atHeight: height do: aBlock

		self atHeight: height listElementDo: [:element |

			aBlock value: element value]



Method do: could finally be implemented trivially as follows. A similar approach was used for keysAndValuesDo:.

 

	do: aBlock

		self atHeight: 1 do: aBlock



We also implemented printOn: for each of the classes in a straigthforward manner. For the list itself, we actually output the sublists at each height. This was useful for debugging but it might be best to output only the height 1 list in the final version. With these changes, we could now execute the following example.



	example2

		"SkipList example2"

		"SkipList example2 at: 50"

		"SkipList example2 at: 50 put: 'Smile'; yourself"

		| list |

		list := SkipList new: 50.

		#('seven' 'two' 'four' 'one' 'five' 'three') do: [:count |

			list at: count put: count size].

		^list



One run of the method resulted in the following output. Since skip list items are created with random heights, we rarely get the same shape for the list.



(3: 'three'=>5)

(2: 'five'=>4 'three'=>5)

(1: 

'five'=>4 'four'=>4 

'one'=>3 'seven'=>5 

'three'=>5 'two'=>3) 

Conclusions

Because of the additional overhead imposed by the skip list organization, traditional sorted lists are much faster for short lists. But skip lists are guaranteed to be faster in the limit. I guess we need to do a few experiments to find out where that break point is.

Listing 1 The ported Wenger SkipList classes.



class:								SkipListElement

superclass:						Object 

instance variables:		key value nexts



class methods

creation

level: anInteger key: aMagnitude value: anObject

	^self new level: anInteger key: aMagnitude value: anObject



instance methods

initializing

level: anInteger key: aMagnitude value: anObject

	self 

		nexts: (Array new: anInteger); 

		key: aMagnitude; 

		value: anObject



get/set

nexts

nexts: skipListElementCollection

key

key: aMagnitude

value

value: anObject

	… code not shown …

   



class:								SkipList

superclass:						Object 

instance variables:		level maximumLevel probability header



class methods

creation

new: initialSize

	"See Bill Pugh, Dept of Computer Science, Univ. of Maryland, paper at ftp://ftp.cs.umd.edu/pub/skipLists"

	| initialProbability initialLevel |

	initialProbability := 0.25.

	initialLevel := (initialSize log: 10) / 

		((1.0 / initialProbability) log: 10).

	^self

		probability: initialProbability

		maximumLevel: (1 max: initialLevel  ceiling)

   

probability: aFloat maximumLevel: anInteger

	^self new probability: aFloat maximumLevel: anInteger



instance methods

initializing

probability: aFloat maximumLevel: anInteger

	| header trailer |

	header := SkipListElement level: anInteger key: 0 value: nil.

	trailer := SkipListElement 

		level: anInteger key: 10000000000 value: nil.

	self

		probability: aFloat;

		level: 1;

		maximumLevel: anInteger;

		header: header.

	1 to: self maximumLevel do: [:currentLevel |

		header nexts at: currentLevel put: trailer]



get/set

level

level: anInteger

maximumLevel

maximumLevel: anInteger

header

header: aSkipListElement

probability

probability: aFloat

	… code not shown …



searching

atOrPast: aMagnitude

	"The first skip list element whose key is >= aMagnitude."

	… see similar method in Listing 2 …

		…

		[aMagnitude > next key] whileTrue: [

		…



search: aMagnitude

	| element |

	element := self atOrPast: aMagnitude.

	element key = aMagnitude

		ifTrue: [^element value]

		ifFalse: [^nil].



inserting

predecessorsOf: aMagnitude

	"All skip list elements for each height (header allowed)."

	… see similar method in Listing 2 …

		…

		[aMagnitude > next key] whileTrue: [

		…

  

insert: aMagnitude value: anObject

	| predecessors element newLevel after |

	predecessors := self predecessorsOf: aMagnitude.

	element := predecessors first nexts at: 1.

	(element key = aMagnitude) ifTrue: [

		element value: anObject. ^self].

	"Obtain a random level for insertion."

	newLevel := self randomLevel.

	newLevel > self level ifTrue: [

		self level + 1 to: newLevel do: [:index |

			predecessors at: index put: self header].

		self level: newLevel].

	"Insert element after predecessors elements."

	element := SkipListElement 

		level: newLevel key: aMagnitude value: anObject.

	1 to: newLevel do: [:index |

		after := (predecessors at: index) nexts at: index.

		(predecessors at: index) nexts at: index put: element.

		element nexts at: index put: after]



deleting  

remove: aMagnitude

	| predecessors element newLevel after |

	predecessors := self predecessorsOf: aMagnitude.

	element := predecessors first nexts at: 1.

	element key = aMagnitude ifFalse: [

		self error: 'key not found'].

	"Link around this element."

	1 to: self level do: [:index |

		after := (predecessors at: index) nexts at: index.

		after == element ifTrue: [

			(predecessors at: index) nexts 

					at: index put: (element nexts at: index)]].

	"Eliminate unnecessary levels."

	self level to: 2 by: -1 do: [:index |

		(self header nexts at: index) key = 10000000000 

				ifFalse: [^self].

		self level: index]



random

randomLevel

	| myProbability |

	myProbability := (100.0 * self probability) truncated.

	1 to: self maximumLevel do: [:index |

		(Random from0To: 99) >= myProbability

			ifTrue: [^index]].

	^self maximumLevel







Listing 2 The new SkipList classes.



class:								SkipList

superclass:						Object 

instance variables:		 height probability header trailer



class methods

creation

new: initialSize

	"See Bill Pugh, Dept of Computer Science, Univ. of Maryland, paper at ftp://ftp.cs.umd.edu/pub/skipLists"

	| initialProbability initialHeight |

	initialProbability := 0.25.

	initialHeight := (initialSize log: 10) / 

		((1.0 / initialProbability) log: 10).

	^self

		probability: initialProbability

		maximumHeight: (1 max: initialHeight ceiling)



probability: aFloat maximumHeight: anInteger

	^self new probability: aFloat maximumHeight: anInteger

   

instance methods

initializing

probability: aFloat maximumHeight: anInteger

	header := SkipListHeader height: anInteger.

	trailer := SkipListTrailer height: anInteger.

	1 to: anInteger do: [:height | 

		header nextAt: height put: trailer].

	self probability: aFloat; height: 1 "of empty list"



get/set

probability

probability: aFloat

height

height: anInteger

	… code not shown …



querying 

maximumHeight

	^header height



sequencing

atHeight: height listElementDo: aBlock

	"Sequence through the list at the given height and supply the skip list elements."

	| current next |

	current := header nextAt: height.

	next := current nextAt: height.

	[next notNil] whileTrue: [

		aBlock value: current.

		current := next. next := current nextAt: height].



atHeight: height do: aBlock

	self atHeight: height listElementDo: [:element |

		aBlock value: element value]



atHeight: height keysAndValuesDo: aBlock

	self atHeight: height listElementDo: [:element |

		aBlock value: element key value: element value]

 

do: aBlock

	self atHeight: 1 do: aBlock



keysAndValuesDo: aBlock

	self atHeight: 1 keysAndValuesDo: aBlock



searching/inserting/deleting

search: aMagnitude

	^self search: aMagnitude ifAbsent: [^nil]

 

search: aMagnitude ifAbsent: aBlock

	| element |

	element := self atOrPast: aMagnitude.

	element = aMagnitude

		ifTrue: [^element value]

		ifFalse: [^aBlock value].



insert: aMagnitude value: anObject

	| predecessors predecessor next element |

	predecessors := self predecessorsOf: aMagnitude.

	next := predecessors first next.

	(next = aMagnitude) ifTrue: [next value: anObject. ^self].

	"Insert element after predecessors elements."

	element := SkipListElement

		height: self randomHeight

		key: aMagnitude

		value: anObject.

	1 to: element height do: [:height |

		predecessor := predecessors at: height.

		predecessor atHeight: height insertNext: element].

	self height: (self height max: element height)

  

remove: aMagnitude

	self 

		remove: aMagnitude 

		ifAbsent: [self error: 'key not found']



remove: aMagnitude ifAbsent: aBlock

	| predecessors predecessor element |

	predecessors := self predecessorsOf: aMagnitude.

	element := predecessors first next.

	element = aMagnitude ifFalse: [^aBlock value].

	"Link around this element."

	1 to: element height do: [:height |

		predecessor := predecessors at: height.

		predecessor nextAt: height put: (element nextAt: height)].

	"Eliminate unnecessary levels."

	self height to: 2 by: -1 do: [:index |

		(header nextAt: index) == trailer ifFalse: [^self].

		self height: index]



searching support 

atOrPast: aMagnitude

	"The first skip list element whose key is >= aMagnitude."

	| previous next |

	"Move right at the top level as far as possible, then drop down and try to go further. Retain the previous skip list element at each level."

	previous := header.

	self height to: 1 by: -1 do: [:height |

		next := previous nextAt: height.

		[next < aMagnitude "i.e., aMag.. < next"] whileTrue: [

			previous := next. next := previous nextAt: height].

		"Drop down one level and go further."].

	"Return the last element at height 1."

   ^next

  

predecessorsOf: aMagnitude

	"All skip list elements for each height (header allowed)."

	| predecessors previous next |

	"Move right at the top level as far as possible, then drop down and try to go further. Retain the previous skip list element at each level."

	predecessors := (Array new: self maximumHeight) 

		atAllPut: header.

	previous := header.

	self height to: 1 by: -1 do: [:height |

		next := previous nextAt: height.

		[next < aMagnitude "i.e., aMag.. < next"] whileTrue: [

			previous := next. next := previous nextAt: height].

		"Save the previous element."

		predecessors at: height put: previous.

		"Drop down one level and go further."].

	"Return the previous element at all levels."

   ^predecessors



dictionary access 

at: key

	^self at: key ifAbsent: [self error: 'key not found']

at: key ifAbsent: aBlock

	^self search: key ifAbsent: aBlock

at: key put: value

	self insert: key value: value



random

randomHeight

	| myProbability |

	myProbability := (100.0 * self probability) truncated.

	1 to: self maximumHeight do: [:index |

		(Random from0To: 99) >= myProbability

			ifTrue: [^index]].

	^self maximumHeight



printing

printOn: aStream

	self maximumHeight to: 1 by: -1 do: [:height |

		aStream nextPutAll: '('; print: height; nextPutAll: ':'.

		self atHeight: height listElementDo: [:element |

			aStream space; print: element].

		aStream nextPutAll: ') '.]







class:								SkipListElement

superclass:						Object 

instance variables:		key value nexts



class methods

creation

height: anInteger key: aMagnitude value: anObject

	^self new 

		height: anInteger key: aMagnitude value: anObject

 

height: anInteger

	^self height: anInteger key: nil value: nil



instance methods

initializing

height: anInteger key: aMagnitude value: anObject

	nexts := Array new: anInteger.

	self key: aMagnitude; value: anObject



get/set

key

key: aMagnitude

value  

value: anObject

	… code not shown (note: nexts is private) …



querying

height

	^nexts size



next facilities

next

	^self nextAt: 1

nextAt: height

	^nexts at: height

nextAt: height put: aSkipListElement

	^nexts at: height put: aSkipListElement



linking

atHeight: height insertNext: aSkipListElement

	| after |

	after := self nextAt: height.

	self nextAt: height put: aSkipListElement.

	aSkipListElement nextAt: height put: after



comparing (private)  

< aMagnitude

	^self key < aMagnitude

= aMagnitude

	^self key = aMagnitude



printing

printOn: aStream

	aStream print: self key; nextPutAll: '=>'; print: self value









class:								SkipListHeader

superclass:						SkipListElement

instance variables:		“none”



instance methods

comparing (private)   

< aMagnitude

	^true

= aMagnitude

	^false



printing

printOn: aStream

	aStream nextPutAll: 'header'







class:								SkipListTrailer

superclass:						SkipListElement

instance variables:		“none”



instance methods

comparing (private)  

< aMagnitude

	^false

= aMagnitude

	^false

  

printing

printOn: aStream

	aStream nextPutAll: 'trailer'

